# Notes on the programming style in CHAMP

Julien Toulouse
Laboratoire de Chimie Théorique
Université Pierre et Marie Curie et CNRS, 75005 Paris, France
julien.toulouse@upmc.fr

July 3, 2016

## Contents

These notes describe the programming paradigm used in the Fortran 90 files of the program CHAMP. This programming paradigm was invented by the theoretical chemist and legendary programmer François Colonna and was given different names over the years: Open Structured Interfaceable Programming Environment (OSIPE) [1], Deductive Object Programming [2], Implicit Reference to Parameters (IRP) [3], as well as Minimal Explicit Declaration (MED). Even though we describe a Fortran implementation of it, this programming paradigm is not restricted to any particular programming language, and in fact has been implemented in many languages. The particular Fortran implementation that I made in CHAMP was very much inspired by the implementation in the program QMCMOL [4] mainly made by Roland Assaraf.

# 1 Introduction

A Fortran computer program produces *objects*, which are Fortran variables, of any type or dimension (scalar/array), containing quantities that we are after. For example, `psi` may be an object containing the value of a wave function evaluated at some electron coordinates. This object `psi` is constructed from other objects. For example, for the case of a Jastrow $\times$ single determinant wave function, we have `psi = jastrow * determinant` where `jastrow` and `determinant` are the values of the Jastrow factor and of the Slater determinant at the same electron coordinates. The later objects are themselves constructed from yet other objects. For example, `determinant` may be constructed from `orbitals` which is an array containing the values of the orbitals at the considered electron coordinates. And so on.

Clearly, there are *dependencies* between these objects, in the sense that the construction of a given object requires that other objects have already been constructed. In other words, the order in which the objects are constructed is important. Usually, in Fortran programs, these dependencies are *not* made explicit, and making sure that the order of the construction of the objects is correct is left to the programmer. In large codes, this may make difficult the implementation of a new object.

In the programming paradigm used here, the dependencies between the objects are made explicit. The programmer then does not need to take care of the order of construction of the objects. This facilitates the implementation of new objects.

# 2 Basis of the paradigm and implementation

The programming paradigm is best explained on a simple example.

## 2.1 Dependency tree and building subroutines

Consider again the object `psi`, constructed from two objects `jastrow` and `determinant`, themselves constructed from other objects according to the following *dependency tree*:




Note that, strictly speaking, this is not a tree but a graph since the branches are not necessarily disjoint. For example, in the example above, the object `electron_coordinates` connects the two branches. We will nevertheless use the vocabulary of computer trees.

This tree encodes the dependencies between the objects. For example, `psi` depends on both `jastrow` and `determinant`. In turn, `jastrow` depends on `jastrow_parameters` and `electron_coordinates`. And so on. We will say that `psi` is a *child* of `jastrow` and `determinant`, and `jastrow` and `determinant` are the *parents* of `psi`. The objects without any parents (those

that are not boxed: `jastrow_parameters`, `electron_coordinates`, `basis_parameters`, `coefficients`) are the *leaves* of the tree. These generally correspond to input parameters which must be given by the user when running a calculation.

The practical implementation of the dependency tree in the program is explained in the next section. For the moment, let us describe how calculations are done, assuming that the information about the dependency tree is available in the program. Each object which is not a leaf of the tree has a *building subroutine* which constructs it. So, written schematically, we have the five following building subroutines in the program, corresponding to the five boxed objects of the tree:

```
subroutine psi_bld
psi = jastrow * determinant
end subroutine psi_bld

subroutine jastrow_bld
jastrow = f(jastrow_parameters,electron_coordinates)
end subroutine jastrow_bld

subroutine determinant_bld
determinant = f(orbitals)
end subroutine determinant_bld

subroutine orbitals_bld
orbitals = f(basis_functions,coefficients)
end subroutine orbitals_bld

subroutine basis_functions_bld
basis_functions = f(electron_coordinates,basis_parameters)
end subroutine basis_functions_bld
```

Here, to simplify, we have shown the explicit expression of `psi` only, the expressions of the other objects are just written as (complicated) functions $f$.

The most part of the program is made of building subroutines, but they do *not* make *all* the program. Here are examples of subroutines which are *not* building subroutines:

- The subroutines reading the input. This is usually where the leaf objects are created.
- The subroutines writing the output.
- The subroutines containing the main iterative algorithms, such as the Monte Carlo algorithm or the wave-function optimization algorithm.

## 2.2 Implementation of the dependency tree (object_create and object_needed)

In CHAMP, it was chosen to give the data of the dependencies between the objects directly in the building subroutines. The advantage of this is that all the information about a given object is localized in its building subroutine, and not delocalized in different places in the code.

Hence, each building subroutine contains a *header* part where the object created by this building subroutine is indicated by `call object_create('...')` and all the parent objects are

listed by `call object_needed('...')`. Thus, the building subroutines of our simple example actually look like:

```
subroutine psi_bld
if(header) then
   call object_create('psi')
   call object_needed('jastrow')
   call object_needed('determinant')
   return
endif
psi = jastrow * determinant
end subroutine psi_bld

subroutine jastrow_bld
if(header) then
   call object_create('jastrow')
   call object_needed('jastrow_parameters')
   call object_needed('electron_coordinates')
   return
endif
jastrow = f(jastrow_parameters,electron_coordinates)
end subroutine jastrow_bld

subroutine determinant_bld
if(header) then
   call object_create('determinant')
   call object_needed('orbitals')
   return
endif
determinant = f(orbitals)
end subroutine determinant_bld

subroutine orbitals_bld
if(header) then
   call object_create('orbitals')
   call object_needed('basis_functions')
   call object_needed('coefficients')
   return
endif
orbitals = f(basis_functions,coefficients)
end subroutine orbitals_bld

subroutine basis_functions_bld
if(header) then
   call object_create('basis_functions')
   call object_needed('electron_coordinates')
   call object_needed('basis_parameters')
   return
endif
basis_functions = f(electron_coordinates,basis_parameters)
```

```
end subroutine basis_functions_bld
```

For all the building subroutines, the header part is executed only once at the beginning of the execution of the program in order to construct the dependency tree. Then, in the actual calculations we always have `header=.false.` so the header part is bypassed.

### 2.3 Manipulating objects (object_provide and object_modified)

All subroutines can manipulate the objects of the dependency tree with the two main commands explained here: `call object_provide` and `call object_modified`.

For example, suppose that the programmer wants to print out the value of `psi` at some place in the code (outside the building subroutines). He just needs to write the following lines:

```
call object_provide('psi')
write(6,*) 'psi=',psi
```

The instruction `call object_provide('psi')` does the following:
It checks if the object `psi` is *valid*, i.e. if it has already been calculated and can be used.

- If yes, then nothing is done.
- If no, then the program checks if its parents `jastrow` and `determinant` are valid.
  - If both are valid, then the program calls the building subroutine `psi_bld` and marks `psi` as valid.
  - If, for instance, only `determinant` is not valid, then the program checks if its parent `orbitals` is valid. If `orbitals` is valid, then the program calls the building subroutine `determinant_bld`, marks `determinant` as valid, then calls the building subroutine `psi_bld` and marks `psi` as valid. If `orbitals` is not valid, the programs checks its parents, and so on.

In other words, `object_provide('psi')` goes down recursively the dependency tree under `psi` until it finds valid objects. It then climbs up the dependency tree, constructing the objects one after the other, in the correct order, until it finally constructs `psi`.

In the case where `object_provide('psi')` goes down to a leaf object of the dependency tree (for example, `basis_parameters`) which is *not* valid, then it gives an error message indicating that this leaf object is necessary to construct `psi` but does not know how to construct this leaf object. Indeed, since a leaf object does not have a building subroutine, the program does not know how to construct it. Leaf objects are instead usually read in from the input file at the beginning of execution and marked as valid then.

In summary, `call object_provide('psi')` has several advantages:

- It is *simple*. The programmer just needs to know the name of the object that he wants, here 'psi'. He does not need to know how this object is calculated by the program. He does not need to know about intermediate objects such as `orbitals`.
- It is *safe*. If a necessary leaf object is not available, the program will properly stops and explains what is missing.
- It is *efficient*, in the sense that this mechanism ensures that only what is needed is calculated, nothing more.

Another important ingredient remains to be explained. What if the value of an object, say `electron_coordinates`, is modified? Then, we need to make sure that if we need any child or grandchild object of `electron_coordinates`, this object must be recalculated with the new value of `electron_coordinates`. This is done by inserting, just after modifying `electron_coordinates`, the instruction `call object_modified('electron_coordinates')`:

```
electron_coordinates = (-2.1, 0.7, 1.5)
call object_modified('electron_coordinates')
```

The instruction `call object_modified('electron_coordinates')` marks `electron_coordinates` as valid, and recursively climbs up the dependency tree to mark as *invalid* all the objects depending on `electron_coordinates`, namely `jastrow`, `basis_functions`, `orbitals`, `determinant`, `psi`.

Thus, if any one of these objects is asked for afterwards, it will be recalculated with the new value of `electron_coordinates`. This is a *safe* mechanism since it prevents the programmer from forgetting to update objects in an iterative algorithm, a frequent bug otherwise.

Note that a typical use of `call object_modified` is after reading in leaf objects from the input to mark them as valid.

# 3 Interface with the non-MED part

The objects created in the non-MED files do not have building subroutines. If they are not available, the program does not know how to construct them. Nevertheless, to facilitate their use in the MED part, they can be added as leaf objects of the dependency tree. For this, we just need to add `call object_modified('...')`, for example in the line just after a non-MED object has been constructed. This provides a safe mechanism for using them, making sure that we use them only when they have been already calculated. In fact, I found this extremely useful when I started to program in CHAMP.

When this is judged useful, a non-MED object can be converted to a MED object, i.e. we write a proper building subroutine for it and it is then a normal object of the dependency tree. The non-MED part of the code can then be progressively evolved in the style of the MED part.

# 4 More details

## 4.1 A building subroutine can create more than one object

Often, it is convenient to construct in the same building subroutine several objects. A given building subroutine thus generally creates several objects. For example, if the building subroutine `jastrow_bld` creates the two object `jastrow` and `jastrow_gradient`, it will look like:

```
subroutine jastrow_bld
if(header) then
   call object_create('jastrow')
   call object_create('jastrow_gradient')
   call object_needed('jastrow_parameters')
   call object_needed('electron_coordinates')
   return
endif
jastrow = f(jastrow_parameters,electron_coordinates)
```

```
jastrow_gradient = g(jastrow_parameters,electron_coordinates)
end subroutine jastrow_bld
```

With this extension, the nodes of the dependency tree are no longer the objects but instead the building subroutines, each one creating a list of objects. This extension does not cause any problem.

### 4.2 Dynamic allocations inside the building subroutines

If an object is an array, it is dynamically allocated inside its building subroutine, using `call object_alloc`. For example:

```
call object_alloc('orbitals', orbitals, norb)
do i=1,norb
   orbitals(i) = ...
enddo
```

Note that the dimension `norb` itself is usually an object handled by the dependency tree. If it changes during the calculation, the array will be reallocated with the new size.

### 4.3 Indexes of objects

The functions `object_modified('...')` and `object_provide('...')` take as argument the name of an object given as a string. The name of the object is then looked up in an array which can take some time. If these functions are called in a part of the code that is very often executed, this is not efficient. In this case, we use directly the index of the object in the array, instead of its name. For example:

```
call object_modified_by_index(electron_coordinates_index)

call object_provide_by_index(psi_index)
```

## 5 Programming Manual

### 5.1 What happens at the beginning of a run?

### 5.2 Definitions

In Fortran, we cannot store the name of an object in a variable : once the code is compiled, the name is lost. If we want to operate on dependencies, we need to retain the names, i.e. we need to have a mapping between each name and a variable. In the CHAMP's implementation of the MED paradigm, there are Fortran objects, and they in general have a string `'name'` and an integer `'index'`.

Note that almost all basic MED instructions are given with the `'name'` variable. When a `'name'` is given, the MED machinery searches for the corresponding `'index'` ; in the core machinery, everything is done with the indexes. Since the search can be costly, as stated before, for repetively used Fortran objects, it is possible to directly give the `'index'`.

In the following sections, we give a glossary of basic and more advanced MED commands implemented in CHAMP.

### 5.2.1 Types

To well understand the commands described in the next sections, one first need to understand how informations about the objects and nodes are stored.

**'object'** (usage: `'object'`)

The informations on an object are stored into a type `object`. All the objects are accumulated in an array `objects` of type `object`, and the object index is the position of an object in that array. The type `object` contains:

| | | | |
|---|---|---|---|
| `character` | :: | `name` | its name |
| `logical` | :: | `associated` | is it associated? |
| `character` | :: | `type` | its type |
| `integer, allocatable` | :: | `dimensions (:)` | its dimensions |
| `logical` | :: | `walkers = .false.` | |
| `logical` | :: | `unweighted = .false.` | |
| `logical` | :: | `valid` | is it valid? |
| `logical` | :: | `freezed = .false.` | is it freezed? |
| `logical` | :: | `object_depend_valid = .false.` | |
| `integer` | :: | `node_create_index = 0` | its building routine |
| `integer, allocatable` | :: | `nodes_depend_index (:)` | |
| `pointer` | :: | `pointer_[type]` | |
| `logical` | :: | `saved = .false.` | |
| `real` | :: | `save_[type]` | |
| (and statistical stuff) | :: | | |

**'node'** (usage: `'node'`)

The informations on a node (a building routine) are stored into a type `node`. All the nodes are accumulated in an array `nodes` of type `node`. The type `node` contains:

| | | | |
|---|---|---|---|
| `character` | :: | `routine_name` | its name |
| `integer` | :: | `routine_address` | its address |
| `integer, allocatable` | :: | `objects_needed_index(:)` | the list of needed objects |
| `integer, allocatable` | :: | `objects_create_index(:)` | the list of created objects |
| `logical` | :: | `valid` | is it valid? |
| `logical` | :: | `debug = .false.` | |
| `logical` | :: | `entered = .false.` | |
| `integer` | :: | `calls_nb = 0` | stat data |
| `real` | :: | `cpu_duration = 0.d0` | stat data |

Hence, for example:

```
nodes(objects(object_index)%node_create_index)%routine_name
```

will give you the `routine_name` of the routine whose index is stored in `node_create_index` of the current object (known by its `object_index`): this is the name of the routine creating the object of index `object_index` !

### 5.2.2 Basics

In this section, we describe in a glossary manner the basic commands of the MED implementation in CHAMP. Those are the commands that are to be called when writting new code or modifying code. In particular they can be found in all the files in CHAMP, by contrast to the "deeper

machinery" commands described in the next section, that are (almost) only called in the "`MED_-tools/*f90`" files themselves.

**object_create** (usage: `[sub] object_create('name'[,'index'])`)

Catalogs an object known by its `'name'` with the current building node index, and possibly returns its (newly produced) `'index'`.
This uses `object_add_once_and_index`.

In details:

- register that the current node creates the object
  (`objects(index)%node_create_index = node_current_index`)
- and that the object is created by the current node
  (add `index` to `nodes(node_current_index)%objects_create_index`)

**object_needed** (usage: `[sub] object_needed('name')`)

Catalogs a dependency of the current building node.
This uses `object_add_once_and_index` (in case the object itself is not already catalogued).

In details:

- register that the object is needed by the current node
  (add `object_index` to `nodes(node_current_index)%objects_needed_index`)
- and that the current node depends on the object
  (add `node_current_index` to `objects(object_index)%nodes_depend_index`)

**object_provide** (usage: `[sub] object_provide('name')`)

Provides an object known by its `'name'` (actually, see `object_provide_by_index`).
This uses `object_index_or_die`.

**object_provide_by_index** (usage: `[sub] object_provide_by_index('index')`)

Provides an object known by its 'index'. The objects needs either to be valid or to have a catalogued building node to be called.

In details:

- if valid, return
- else execute the building node
  (`call node_exe_by_index(objects(index)%node_create_index)`)

**object_provide_in_node** (usage: `object_provide_in_node`)

**object_provide_in_node_by_index** (usage: `object_provide_in_node_by_index`)

**object_modified** (usage: `[sub] object_modified('name')`)

Validate an object known by its `'name'` and invalidate all children (known by their `'index'`), actually, see : `object_modified_by_index`.
This uses `object_add_once_and_index`.

**object_modified_by_index** (usage: `[sub] object_modified_by_index('index')`)

Validate an object known by its `'index'` and invalidate all children (known by their `'index'`).

In details:

- validate the object
  (`objects(index)%valid = .true.`)
- invalidate all children created by nodes depending on the object
  (This uses `object_invalidate_by_index` on all `nodes(objects(index)%nodes_depend_index(node_i))%objects_create_index` for all the `node_i` in `objects(index)%nodes_depend_index`.)

**object_alloc** (usage: `[sub] object_alloc('name','object'[,dimensions])`)

Allocate (or: reallocate) and associate (or: reassociate) an object.
This uses `alloc` and `object_associate`.

**object_write** (usage: `object_write('name')`)

**object_write_2** (usage: `object_write_2('name')`)

**object_write_by_index** (usage: `object_write_by_index('index')`)

**object_save** (usage: `object_save('name')`)

**object_restore** (usage: `object_restore('name')`)

### 5.2.3 Deeper machinery

In this section are described in a glossary manner the more core commands of the MED implementation in CHAMP.

**object_index** (usage: `[fun] object_index('name')`)

Searches through all objects to find the `'index'` corresponding to `'name'`. Returns the index or 0 if object `'name'` is not catalogued.

**object_index_or_die** (usage: `[fun] object_index_or_die('name')`)

Same as `object_index`, but will die if the object is not catalogued (if `object_index` return 0).

**object_add** (usage: `[sub] object_add('name')`)

Catalog a new object.
This means adding an entry to the `'objects'` array, with `'name'` as name and default values for other data

**object_add_and_index** (usage: `[sub] object_add_and_index('index')`)

Catalogs a new object that has no name and returns its (newly produced) index (actually, see: `object_add_once_and_index`) .

**object_add_once_and_index** (usage: `[sub] object_add_once_and_index('name','index')`)

Returns the `'index'` corresponding to a `'name'`. Assigns a new one if needed, i.e. catalogs the new object (actually, see: `object_add`). This is used by all MED front-end routines.

**object_valid** (usage: `[fun] object_valid('name')`)

Returns whether the object is valid or not (actually, see `object_valid_by_index`).
This uses `object_add_once_and_index`.

**object_valid_by_index** (usage: `[fun] object_valid_by_index('index')`)

Returns the value of `objects(index)%valid`

**object_valid_or_die** (usage: `[sub] object_valid_or_die('name')`)

Dies if object not valid (uses `object_valid`)

**object_depend_valid_by_index** (usage: `object_depend_valid_by_index`)

**object_invalidate** (usage: `[sub] object_invalidate('name')`)

Invalidates an object known by its `'name'`, actually: see `object_invalidate_by_index`.
This uses `object_add_once_and_index`.

**object_invalidate_by_index** (usage: `[sub] object_invalidate_by_index('name')`)

Invalidates an object known by its `'index'` and all its children (known by their `'index'`).

In details:

- if all are already invalid, return
- if object is freezed, return
- invalidate the object and its building node
  (i.e. `objects(object_index)%valid = .false.` and `nodes(objects(object_index)%node_create_index)%valid = .false.`
- invalidate all objects created by nodes depending on the object
  (This uses `object_invalidate_by_index` itself on all the object indexes in `nodes(objects(object_index)%nodes_depend_index(node_i))%objects_create_index` for all `node_i` in `objects(object_index)%nodes_depend_index`.)

**object_associate** (usage: `[sub] object_associate('name','object'[,dimensions])`)

Associate a pointer to an object
This uses `object_add_once_and_index`.
In details:
HERE

**object_deassociate** (usage: `[sub] object_deassociate('name')`)

Deassociate a pointer and an object.
This uses `object_add_once_and_index`.
In details:
HERE

**object_associated_or_die_by_index** (usage: `[sub] object_associated_or_die_by_index('index')`)

Dies if the object is not associated (i.e. if `objects(index)%associated` is `false`).

**object_alloc_by_index** (usage: `[sub] object_alloc_by_index('index')`)

Allocate pointer to an object.

**object_release** (usage: `[sub] object_release('name','object')`)

Deallocate and deassociate an object
This uses `release` and `object_deassociate`.

**node_exe** (usage: `[sub] node_exe('name')`)

Executes a node known by its `'name'` (actually: see `node_exe_by_index`).
This uses `node_index`.

**node_exe_by_index** (usage: `[sub] node_exe_by_index('index')`)

Executes a node known by its `'index'`.

In details:

- if node is valid, return
- provide all the needed objects of the node
  (This uses `object_provide_from_node_by_index` on all `nodes(node_index)%objects_needed_index`.)
- execute the current node.
  (This uses `exe_by_address_0`, a C routine.)
- validate the current node and all its created objects
  (i.e. all objects in `nodes(node_index)%objects_create_index` are set to be valid).

**object_write_no_routine_name** (usage: `object_write_no_routine_name('name')`)

**object_write2_no_routine_name** (usage: `object_write2_no_routine_name('name')`)

**object_restore_by_index** (usage: `object_restore_by_index('index')`)

**object_freeze** (usage: `object_freeze('name')`)

**object_zero** (usage: `object_zero('name')`)

**object_provide_from_node_by_index** (usage: `object_provide_from_node_by_index`)

**object_modfied2_by_index** (usage: `object_modfied2_by_index`)

### 5.2.4 Simple wrappers (not MED machinery)

**alloc** (usage: `alloc`)

alloc=allocate

**release** (usage: `release`)

release=deallocate

**alloc_range** (usage: `alloc_range`)

**alloc_test** (usage: `alloc_test`)

#### 5.2.5 Averages

**object_average_request** (usage: `object_average_request('name')`)

**object_average_define** (usage: `object_average_define('name','name_av')`)

### 5.3 About wavefunction and derivatives

Consider

$\Phi = \sum c_I |C_I\rangle = \sum c_I \sum c_{k_I} |k_I\rangle = \sum c_I \sum c_{k_I} |\uparrow (k_I)\rangle |\downarrow (k_I)\rangle$,

and the excitation of a determinant:

$\hat{E}_{kl} (|\uparrow (k_I)\rangle |\downarrow (k_I)\rangle) = \hat{E}_{kl} (|\uparrow (k_I)\rangle) |\downarrow (k_I)\rangle + |\uparrow (k_I)\rangle \hat{E}_{kl} (|\downarrow (k_I)\rangle)$ .

We have the following things:

| | | |
|---|---|---|
| $c_I$ | :: | `csf_coef(:)` |
| $c_{k_I}$ | :: | `cdet_in_csf(:,`$c_I$`)` |
| $\|k_I\rangle$ | :: | `iwdet_in_csf(:,`$c_I$`)` |
| from $\|k_I\rangle$ to $\|\uparrow (k_I)\rangle$ | :: | `det_to_det_unq_up(#)` |
| from $\|k_I\rangle$ to $\|\downarrow (k_I)\rangle$ | :: | `det_to_det_unq_dn(#)` |

Note that `ex_orb_ind` maps the orbital parameters to the single excitations. Indeed, for act-act with orthonormality imposed, there are subtleties: some single excitation are not free orbital parameters but reverse excitations stored in `ex_orb_ind_rev`. Otherwise, we have `ex_orb_ind=Id` and `ex_orb_ind_inv=0`.

## References


[1] F. Colonna, L.-H. Jolly, R. A. Poirier, J. G. Ángyán and G. Jansen, Comp. Phys. Comm. **81**, 293 (1994).

[2] F. Colonna, `http://arxiv.org/abs/cs/0601035v1`.

[3] A. Scemama, `http://arxiv.org/abs/0909.5012v1`.

[4] QMCMOL, a quantum Monte Carlo program written by R. Assaraf, F. Colonna, X. Krokidis, P. Reinhardt and coworkers.